Abrupt Changes and the Dirac Delta Function

```
> restart;with(inttrans);
```

$$[addtable, fourier, fouriercos, fouriersin, hankel, hilbert, invfourier, invhilbert, invlaplace, invmellin, laplace, mellin, savetable]$$

Now let us use Laplace transforms step-by-step to solve the following IVP where N is a constant.

```
> DE:=diff(p(t),t)-(2/25)*p(t)=-N*(Dirac(t-1/2)+Dirac(t-1)+Dirac(t-3/2)+Dirac(t-2));
```

$$DE := \frac{d}{dt} p(t) - \frac{2}{25} p(t) = -N\left(\mathrm{Dirac}\left(t-\frac{1}{2}\right) + \mathrm{Dirac}(t-1) + \mathrm{Dirac}\left(t-\frac{3}{2}\right) + \mathrm{Dirac}(t-2)\right)$$

```
> IC:=p(0)=1600;
```

$$IC := p(0) = 1600$$

We take the Laplace transform of each side of the equation.

```
> L:=laplace(lhs(DE),t,s)=laplace(rhs(DE),t,s);
```

$$L := s\, laplace(p(t), t, s) - p(0) - \frac{2}{25}\, laplace(p(t), t, s) = -N\left(e^{-\frac{1}{2}s} + e^{-s} + e^{-\frac{3}{2}s} + e^{-2s}\right)$$

We substitute in the value of the initial condition.

```
> LI:=subs(p(0)=1600,L);
```

$$LI := s\, laplace(p(t), t, s) - 1600 - \frac{2}{25}\, laplace(p(t), t, s) = -N\left(e^{-\frac{1}{2}s} + e^{-s} + e^{-\frac{3}{2}s} + e^{-2s}\right)$$

We solve for $\boldsymbol{L}\{x\}$, denoting it by the variable **LT**, and then expanding the solution.

```
> LT:=solve(LI,laplace(p(t),t,s));
```

$$LT := -\frac{25\left(N e^{-\frac{1}{2}s} + N e^{-s} + N e^{-\frac{3}{2}s} + N e^{-2s} - 1600\right)}{25s-2}$$

Finally, we take the inverse Laplace transform to find our solution.

```
> soln_p(t):=invlaplace(LT,s,t);
```

$$soln\_p(t) := 1600\, e^{\frac{2}{25}t} + N\left((\mathrm{Heaviside}(2-t)-1)\, e^{\frac{2}{25}t-\frac{4}{25}} + \left(\mathrm{Heaviside}\left(\frac{1}{2}-t\right)-1\right) e^{\frac{2}{25}t-\frac{1}{25}} + (\mathrm{Heaviside}(1-t)-1)\, e^{\frac{2}{25}t-\frac{2}{25}} + \left(\mathrm{Heaviside}\left(\frac{3}{2}-t\right)-1\right) e^{\frac{2}{25}t-\frac{3}{25}}\right)$$

Of course, we could have used the simple Maple way.

```
> infolevel[dsolve]:=5;
```

$$infolevel_{dsolve} := 5$$

```
> dsolve({DE,IC},p(t));
```

```
Methods for first order ODEs:
--- Trying classification methods ---
trying a quadrature
trying 1st order linear
<- 1st order linear successful
```

$$p(t) = \left(-N\left(e^{-\frac{1}{25}}\,\mathrm{Heaviside}\left(t-\frac{1}{2}\right) + e^{-\frac{2}{25}}\,\mathrm{Heaviside}(t-1) + e^{-\frac{3}{25}}\,\mathrm{Heaviside}\left(t-\frac{3}{2}\right) + e^{-\frac{4}{25}}\,\mathrm{Heaviside}(t-2)\right) + 1600\right) e^{\frac{2}{25}t}$$

```
> dsolve({DE,IC},p(t),method=laplace);
```

```
dsolve/inttrans/solveit: Transform of eqns is {_s1*laplace/internal(p
(t) t _s1)-1600-(2/25)*laplace/internal(p(t) t _s1)+N*(exp(-(1/2)*_s1)+
exp(-_s1)+exp(-(3/2)*_s1)+exp(-2*_s1))}
dsolve/inttrans/solveit: Algebraic Solution is {laplace/internal(p(t) t
_s1) = -25*(N*exp(-_s1)-1600+N*exp(-2*_s1)+N*exp(-(1/2)*_s1)+N*exp(-
(3/2)*_s1))/(25*_s1-2)}
```

$$p(t) = 1600\,e^{\frac{2}{25}t} + N\left((\mathrm{Heaviside}(2-t)-1)\,e^{\frac{2}{25}t-\frac{4}{25}} + \left(\mathrm{Heaviside}\left(\frac{1}{2}-t\right)-1\right) e^{\frac{2}{25}t-\frac{1}{25}} + (\mathrm{Heaviside}(1-t)-1)\,e^{\frac{2}{25}t-\frac{2}{25}} + \left(\mathrm{Heaviside}\left(\frac{3}{2}-t\right)-1\right) e^{\frac{2}{25}t-\frac{3}{25}}\right)$$

Finally, assuming that we know also that $p(2) = 0$, we solve for N. We actually use 2.00001 so as to make sure we include the impulse at $t = 2$ since the unit step function $u(t-2)$ is undefined there.

```
> soln_p(2):=subs(t=2.00001,soln_p(t));
```

$$soln\_p(2) := 1600\,e^{0.1600008000} + N\left((\mathrm{Heaviside}(-0.00001)-1)\,e^{8.000\,10^{-7}} + (\mathrm{Heaviside}(-1.500010000)-1)\,e^{0.1200008000} + (\mathrm{Heaviside}(-1.00001)-1)\,e^{0.08000080000} + (\mathrm{Heaviside}(-0.500010000)-1)\,e^{0.0400008000}\right)$$

```
> N:=round(solve(soln_p(2)=0,N));
```

$$N := 442$$

Finally, let's put this value of N into our solution and then plot the solution over a three year time period.

```
> plot(soln_p(t),t=0..3,discont=true, thickness=3);
```


1600
1400
1200
1000
800
600
400
200
0
0
1
2
3
t
Curve 1